

Overview of Physical Activity Variables in the ALSWH

Physical activity (PA) is measured by metabolic equivalent (MET). This is the unit used to define levels of activity, in multiples of resting metabolic rate. One MET is defined as energy expenditure at rest, usually equivalent to 3.5mL of oxygen uptake per kg per minute.

There are two PA variables in the ALSWH data that attempt to capture MET. The two variables are ExStat and MetMin. They differ only in the weightings given to the number of physically active minutes and in the cut-offs used in their categorisation. They are each categorised into four groups; ExStat's categorised variable is ExGrp and MetMin's is MetMinExGrp.

The ALSWH questionnaires ask the time (hours, minutes) spent last week

Walking briskly

Moderate leisure activity

Vigorous leisure activity

These times are weighted with the numbers in Table 1 to calculate the MET variables. The actual algorithm is more complex than the above statement would suggest and the SAS code is given below in the Appendix.

Table 1: Weightings and Cut-Offs used

	ExStat	MetMin
	Weightings	
Activities		
Walking Briskly	3.0	3.33
Moderate leisure activity	4.0	3.33
Vigorous leisure activity	7.5	6.66
Categories	Cut-Offs	
	ExGrp	MetMinGrp
1 Nil/sedentary	0 - <40	0 – <33.3
2 Low	40 – <600	33.3 – <500
3 Moderate	600 – <1200	500 – <1000
4 High	1200 +	1000 +

Historically the variable ExStat was first used on ALSWH, (Brown et al, 2008). Later the variable MetMin was added to all relevant surveys, (Brown et al, 2012; Brown & Pavey, 2016). The 1989-95 cohort only has the MetMin, MetMinGrp variables.

1921 – 1926, 1946-51, and 1973-78 cohorts

Survey 1

The first surveys, Survey 1, for the three cohorts have the variable `exstat_old`. This is not directly equivalent to `exstat`. See Data Dictionary Supplement 2.5, Physical Activity – Survey 1

Surveys 2 onwards

Surveys 2 onwards have the four variables in Table 1. Note that the Six Month Follow up survey for 1921-26 does not have any PA variables. Also, the PA variable in Survey 2 in the 1946-51 cohort has some problems and we recommend that it not be used in longitudinal analysis with the other waves. See the Recommendation in the Data Dictionary Supplement Section 2.5 , **Physical Activity – Survey 2 and later surveys**

1989-95 Cohort

The earlier version of the PA variable, `ExStat`, is not included in this cohort. The grouping variable is called both `ExGrp` and `MetMinGrp` – these are the same variable.

References

Brown, W. J., Burton, N. W., Marshall, A. L., & Miller, Y. D. (2008). Reliability and validity of a modified self-administered version of the Active Australia physical activity survey in a sample of mid-age women. *Australian and New Zealand Journal of Public Health*, 32(6), 535-541.
doi:10.1111/j.1753-6405.2008.00305.x

Brown, W.J., Bauman, A., Bull, F., & Burton, N.W. (2013). Development of Evidence-based Physical Activity Recommendations for Adults (18-64 years). Report prepared for the Australian Government Department of Health. Canberra, Australia: Australian Department of Health.

Brown, W. J., & Pavey, T. (2016). Physical Activity in Mid-Age and Older Women: Lessons from the Australian Longitudinal Study on Women’s Health. *Kinesiology Review*, 5(1), 87-97.
doi:10.1123/kr.2015-0057

Appendix

SAS code for Physical Activity used in Surveys 2 onwards

This example is from the second survey of the 1973-78 cohort. (Young 2)

```
*** First recode missing values using standard logical methods ***  
*** Allocate exercise status variable - y2exstat  
      (metmins)
```

```

array leisure{3}  y2q65a  y2q65b  y2q65c  ;
array numtimes{4} y2q65a  y2q65b  y2q65c  y2q65d  ;

array hours{4}    y2q66ah  y2q66bh  y2q66ch  y2q66dh  ;
array minutes{4} y2q66am  y2q66bm  y2q66cm  y2q66dm  ;

array totmins{3} totmina  totminb  totminc  ;

/*****
# times doing leisure activities

If some items in this group have a nonzero response AND
no items in this group have a response of 'zero'
THEN code missing values to zero.
*****/

        if nmiss (of numtimes{*}) in (0,4) then ;
else if y2q65a=0 or y2q65b=0 or y2q65c=0 or y2q65d=0 then ;
else do i = 1 to 4 ;
        if numtimes{i}=.  then numtimes{i}=0 ;
end ;

/*****

hours & minutes spent doing leisure activities

If hours spent doing activities is not missing AND
the corresponding minutes are missing

```

THEN code missing values in minutes to 0.

(Similarly, for minutes not missing and hours missing)

```
*****/
do i = 1 to 4 ;
    if hours{i} ne . and minutes{i} = . then minutes{i} = 0
;
    if hours{i} = . and minutes{i} ne . then hours{i} = 0 ;
end ;

/*****
***
*** If number of times=0 and hours=. and mins=. ***
*** then set hours and mins to 0. ***
***
*** If hours=0 and mins=0 and number of times=. ***
*** then set number of times to 0. ***
***
*****/
do a=1 to 4 ;
    if numtimes(a)=0 and hours(a)=. and minutes(a)=. then do ;
        hours(a)=0 ;
        minutes(a)=0 ;
    end ;
end ;

do a=1 to 4 ;
```

```

        if numtimes(a)=. and hours(a)=0 and minutes(a)=0 then
numtimes(a)=0 ;
end ;

/*****

***                                     ***

***   If hours>0 and mins>0 and number of times=.   ***

***   then set number of times to 0.                 ***

***                                     ***

*****/

        do a=1 to 4 ;

                if numtimes(a)=. and 0<=hours(a)<=99 and
0<=minutes(a)<=90 then numtimes(a)=0 ;

        end ;

/*****

        Calculate total # times leisure activity reported

*****/

        y2q65tot = y2q65a + y2q65b + y2q65c ;

/*****

        Re-assign extreme values for the # times reported doing

        leisure activities, that is totals > 56 times

*****/

if y2q65tot >56 then do i = 1 to 3 ;

        leisure{i} = round ( ((56/y2q65tot) * leisure{i}), 1) ;

```

```

end ;

else if y2q65tot = . then do ;
    if (y2q65a+y2q65b) >56 then do ;
        y2q65a = round ( ( (56/(y2q65a+y2q65b) ) * y2q65a), 1) ;
        y2q65b = round ( ( (56/(y2q65a+y2q65b) ) * y2q65b), 1) ;
        end ;
    else if (y2q65a+y2q65c) > 56 then do ;
        y2q65a = round ( ( (56/(y2q65a+y2q65c) ) * y2q65a), 1) ;
        y2q65c = round ( ( (56/(y2q65a+y2q65c) ) * y2q65c), 1) ;
        end ;
    else if (y2q65b+y2q65c) >56 then do ;
        y2q65b = round ( ( (56/(y2q65b+y2q65c) ) * y2q65b), 1) ;
        y2q65c = round ( ( (56/(y2q65b+y2q65c) ) * y2q65c), 1) ;
        end ;
    else do i = 1 to 3 ;
        if leisure{i} >56 then leisure{i}=56 ;
        end ;
    end ;

if y2q65d>56 then y2q65d=56 ;

/*****
Calculate the total minutes doing each leisure activity
*****/

do i = 1 to 3 ;
    totmins{i} = (60*hours{i}) + minutes{i} ;

end ;

```

```

/*****

Calculate total minutes of leisure activity reported
*****/

y2q66min = round ((totmina + totminb + totminc), .01);

/*****

Re-assign extreme values for the total time reported doing
leisure activities, that is totals > 40 hrs (>2400 minutes)
*****/

if y2q66min >2400 then do i = 1 to 3 ;
    totmins{i} = round ( ((2400/y2q66min) * totmins{i}), .01) ;
end ;

else if y2q66min = . then do ;
    if (totmina + totminb) >2400 then do ;
        totmina = round ( ( (2400/(totmina + totminb) ) *
totmina), .01) ;
        totminb = round ( ( (2400/(totmina + totminb) ) *
totminb), .01) ;
        end ;
    else if (totmina + totminc) > 2400 then do ;
        totmina = round ( ( (2400/(totmina + totminc) ) *
totmina), .01) ;
        totminc = round ( ( (2400/(totmina + totminc) ) *
totminc), .01) ;
        end ;
    else if (totminb + totminc) >2400 then do ;

```

```

        totminb = round ( ( (2400/(totminb + totminc) ) *
totminb), .01) ;

        totminc = round ( ( (2400/(totminb + totminc) ) *
totminc), .01) ;

        end ;

    else do i = 1 to 3 ;

        if totmins{i} >2400 then totmins{i}=2400 ;

        end ;

    end ;

/*****
*

    Create an overall measure of leisure time physical activity
    which is the sum of minutes for each activity, each weighted
    for level of intensity of that activity.

As per Brown et al, 'A new method for estimating adequate physical
activity for people of different ages'

*****/

/

y2exstat = (3*totmina) + (4*totminb) + (7.5*totminc) ;

if y2exstat=. then y2exgrp=. ;

else if y2exstat<37.5 then y2exgrp = 1 ;

else if 37.5<=y2exstat<600 then y2exgrp = 2 ;

else if 600<=y2exstat<1200 then y2exgrp = 3 ;

else if 1200<=y2exstat then y2exgrp = 4 ;

```



```
/******  
  
y2metmin=sum((totmina*3.33),(totminb*3.33),(totminc*6.66));  
  
if 0<=y2metmin<33.3 then y2metminexgrp=1;  
else if 33.3<=y2metmin<500 then y2metminexgrp=2;  
else if 500<=y2metmin <1000 then y2metminexgrp=3;  
else if y2metmin>=1000 then y2metminexgrp=4;  
**label y2metminexgrp=' ALSWH metmin exercise groups';  
  
drop a i y2q65tot totmina totminb totminc y2q66min ;
```